

Tema 7

Càlcul diferencial de funcions reals de variable real (1)

Objectius:

1. Calcular la derivada d'una funció amb wxMaxima.
2. Calcular derivades d'ordre superior.
3. Calcular la diferencial d'una funció.
4. Calcular derivades de funcions definides implícitament.

Continguts:

- 07-1. Derivada d'una funció. Funcions derivables. Càlcul de derivades.
- 07-2. Regles de derivació. Derivades d'ordre superior.
- 07-3. Diferencial d'una funció. Aplicacions.
- 07-4. Funcions implícites. Derivació de funcions implícites.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.
- AP86 APOSTOL, T.M. (1986)
Análisis Matemático
- BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima
- BU07 DE BURGOS, JUAN (2007)
Cálculo Infinitesimal de una variable (segunda edición).

- ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.
- ES02 ESTEP, DONALD (2002)
Practical Analysis in one variable
- GV07a GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 1: números y ecuaciones
- GV07b GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 2: límites y derivadas
- JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima
- RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería
- RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.
- SP95 SPIVAK, MICHAEL (1995)
Calculus (Càlcul Infinitesimal).
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

07-1.- Derivada d'una funció. Funcions derivables. Càlcul de derivades

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_07-1.wxm**.

Si $f : A \subset \mathbb{R} \rightarrow \mathbb{R}$ és una funció real de variable real i $a \in A$ és un punt interior del camp d'existència de la funció, s'anomena derivada de la funció en aquest punt, el següent límit, si existeix

$$Df(a) = \lim_{h \rightarrow 0} \frac{f(a+h) - f(a)}{h} = \lim_{x \rightarrow a} \frac{f(x) - f(a)}{x - a}$$

La interpretació geomètrica d'aquest concepte és prou coneguda: és el pendent de la recta tangent a la gràfica de la funció en el punt $(a, f(a))$ quan aquesta recta existeix, expressada com el límit dels pendents de les rectes secants a aquesta gràfica que passen pels punts $(a, f(a))$.

Per exemple, si es considera la funció definida per:


$$f(x) = x^2 - 2x + 2, \quad x \in \mathbb{R}$$

aleshores a continuació es pot veure aquesta interpretació geomètrica amb unes rectes secants que passen pel punt $(2, f(2))$.

Implementant els càlculs amb wxMaxima per a la representació gràfica, s'obté:

```
(%i1) f1(x):=x^2-2*x+2$
      y1(x):=4*x-6$    y2(x):=3*x-4$    y3(x):=2*x-2$

--> wxplot2d([f1(x),y1(x),y2(x),y3(x)], [x,1,4], [legend, "f1(x)", "y1(x)", "y2(x)", "y3(x)"],
[style, [lines, 2]], [gnuplot_preamble, "set grid"], [nticks, 200])$
```



Amb wxMaxima es pot calcular la derivada d'una funció en un punt aplicant la definició anterior:

```
(%i6) 'limit((f1(x)-f1(2))/(x-2), x, 2)=limit((f1(x)-f1(2))/(x-2), x, 2);
```

```
(%o6)  $\lim_{x \rightarrow 2} \frac{x^2 - 2x}{x - 2} = 2$ 
```


Si una funció té derivada en un punt es diu derivable o diferenciable en aquest punt i es diu no derivable o no diferenciable, en cas contrari. A continuació es mostra un exemple de funció no derivable en un punt.

```
(%i7) f2(x):=abs(x^2-2*x-8);
```

```
(%o7)  $f2(x) := |x^2 - 2x - 8|$ 
```

```
(%i8) wxplot2d([f2(x)], [x,0, 8],[y,-2,12], [legend, "f2(x)"], [ylabel, ""], [style, [lines, 2]], [gnuplot_preamble, "set grid"] )$
```

plot2d: some values were clipped.



En el cas de la funció $f(x) = |x^2 - 2x - 8|$ representada en la gràfica anterior i si es considera el punt $a = 4$, la resposta de wxMaxima és:

```
(%i9) 'limit((f2(x)-f2(4))/(x-4), x, 4)=limit((f2(x)-f2(4))/(x-4), x, 4);
```

```
(%o9)  $\lim_{x \rightarrow 4} \frac{|x^2 - 2x - 8|}{x - 4} = \text{und}$ 
```

La conclusió és que la funció no és derivable en aquest punt. Aleshores es poden calcular les derivades laterals, designades $D_+ f(a)$ i $D_- f(a)$, que són els límits laterals de l'expressió de la definició de derivada d'una funció en un punt. Si la funció és

derivable ambdós límits existeixen i són iguals i si la funció no és derivable, aleshores o bé les derivades laterals existeixen i no són iguals o bé algun dels límits no existeix. En el cas de la funció considera abans, les derivades laterals són:

```
(%i10) '(D-f2(4)=limit((f2(x)-f2(4))/(x-4), x, 4, minus)) =
          limit((f2(x)-f2(4))/(x-4), x, 4, minus);
(%o10) D-f2(4)= lim  $\frac{f_2(x)-f_2(4)}{x-4} = -6$ 
          x->4-
```

```
(%i11) '(D+f2(4)=limit((f2(x)-f2(4))/(x-4), x, 4, plus)) =
          limit((f2(x)-f2(4))/(x-4), x, 4, plus);
(%o11) D+f2(4)= lim  $\frac{f_2(x)-f_2(4)}{x-4} = 6$ 
          x->4+
```

Càlcul de la derivada d'una funció

Per calcular la derivada d'una funció en un punt qualsevol amb wxMaxima, és a dir, per calcular l'anomenada funció derivada, cal aplicar la instrucció “diff” indicant entre parèntesi el símbol assignat a la funció que ha d'haver estat definida prèviament i la variable respecte la qual es deriva. Per exemple, si s'ha definit $f_1(x) = x^2 - 2x + 2$ aleshores:

```
(%i12) Df1(x)=diff(f1(x), x);
(%o12) Df1(x)= 2 x-2
```

Si es vol calcular la derivada en un punt concret, és a dir, el valor de la funció derivada en aquest punt, es podria pensar que la metodologia és similar a la que permet el càlcul del límit d'una funció en un punt. Així, si es vol calcular la derivada de la funció anterior en el punt $a = 3$, es podria pensar que la instrucció a aplicar és:

```
(%i13) Df1(3)=diff(f1(x), x, 3);
(%o13) Df1(3)= 0
```

resultat que no és correcte ja que $Df_1(3) = 4$. Així, doncs, aquesta instrucció no és la correcta per aquesta finalitat i cal implementar una estratègia adequada. Una opció plausible és aplicar la instrucció “define” per definir la funció derivada i aplicar aquesta funció al punt concret. Així, en el exemple anterior, es pot procedir de la manera següent:

```
(%i14) define ( Df1(x) , diff(f1(x), x) );
(%o14) Df1(x):= 2 x-2
```

Ara es pot calcular la derivada en un punt concret o en un punt qualsevol:

```
(%i15) 'Df1(3) = Df1(3); 'Df1(a) = Df1(a);
(%o15) Df1(3)=4
(%o16) Df1(a)=2 a-2
```

Cal anar amb atenció amb aquesta metodologia. Per exemple, si es considera la funció definida per:

$$F(x) = \begin{cases} x^2 \sin\left(\frac{1}{x}\right), & \text{si } x \neq 0 \\ 0, & \text{si } x = 0 \end{cases}$$

i ara es calcula la derivada amb wxMaxima, resulta

```
(%i17) F(x):=x^2*(sin(1/x)); diff(F(x),x);
(%o17) F(x):=x^2 sin(1/x)
(%o18) 2 sin(1/x) x - cos(1/x)
```

Per tant, si es defineix la funció derivada mitjançant:

```
(%i19) define ( DF(x) , diff(F(x),x) );
(%o19) DF(x):= 2 sin(1/x) x - cos(1/x)
```

i ara es vol calcular la derivada a l'origen com el valor en aquest punt de la funció, és clar que això no es pot fer ja que l'expressió anterior no està definida a l'origen. En aquest cas es pot veure que la funció considerada és derivable a l'origen i es compleix $DF(0) = 0$; en efecte:

```
(%i20) 'DF(0)=limit((F(x))/(x), x, 0);
(%o20) DF(0)=0
```


Derivada infinita.

Un altre cas a considerar és la possibilitat que el límit de la definició de derivada no sigui finit i els límits laterals existeixin i coincideixin. Tal és el cas, per exemple de la funció definida per:

$$g(x) = \begin{cases} -\sqrt{-x}, & \text{si } x \leq 0 \\ \sqrt{x}, & \text{si } x > 0 \end{cases}$$

```
(%i21) g1(x):=-sqrt(-x);      g2(x):=sqrt(x);
      g(x):= if x<=0 then g1(x) else g2(x);
(%o21) g1(x):=-sqrt(-x)
(%o22) g2(x):=sqrt(x)
(%o23) g(x):=if x<=0 then g1(x) else g2(x)

(%i24) wxplot2d([g(x)], [x,-2,2], [y, -2,2], [legend, "g1(x)"] , [ylabel, ""], [style, [lines, 2]] ,
      [gnuplot_preamble, "set grid"], [nticks, 200] )$
```



Si es calcula la derivada de cada una de les expressions analítiques de la funció, resulta

```
(%i25) diff(g1(x), x);    diff(g2(x), x);
(%o25) 1/(2*sqrt(-x))
(%o26) 1/(2*sqrt(x))
```

Les derivades de la funció no estan definides a l'origen; per tant s'han de calcular les derivades laterals en aquest punt aplicant la definició de derivada lateral:

```
(%i27) '(D-g(0)) = limit((g1(x)-g1(0))/(x), x, 0, minus);
      '(D+g(0)) = limit((g2(x)-g2(0))/(x), x, 0, plus);
(%o27) D-g(0)=∞
(%o28) D+g(0)=∞
```

En conclusió es pot afirmar que $Dg(0) = +\infty$ (extensió de la derivada a la recta real ampliada).

Per acabar aquest apartat mostrem que amb wxMaxima es pot calcular la funció derivada d'una funció que tingui paràmetres en la seva expressió analítica. Així, per exemple:

```
(%i29) f4(x):=a_0 + a_1*x + a_2*x^2 + a_3*x^3;
      Df4(x) = diff(f4(x), x);
(%o29) f4(x):=a_0+a_1 x+a_2 x^2+a_3 x^3
      (%o30) Df4(x)=3 a_3 x^2+2 a_2 x+a_1
(%i31) f5(x):=exp(a*x);
      Df5(x) = diff(f5(x), x);
(%o31) f5(x):=exp(a x)
      (%o32) Df5(x)=a %ea x
(%i33) f6(x):=sin(a*x)+x*log(b*x);
      Df6(x) = diff(f6(x), x);
(%o33) f6(x):=sin(a x)+x log(b x)
      (%o34) Df6(x)=log(b x)+a cos(a x)+
```

07-2. Regles de derivació. Derivades d'ordre superior.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_07-2.wxm**.

Les regles de derivació, és a dir, la relació entre la derivació i les operacions elementals amb funcions, es poden escriure amb wxMaxima, tal com es mostra a continuació. Observi's que es fa servir la notació clàssica per a la derivada d'una funció i no la més moderna amb l'operador derivació D .

Derivada de la suma i diferència de funcions:

```
(%i1) 'diff(f(x)+g(x), x)=diff(f(x)+g(x), x);  
'diff(f(x)-g(x), x)=diff(f(x)-g(x), x);
```

$$(\%o1) \frac{d}{dx} (g(x) + f(x)) = \frac{d}{dx} g(x) + \frac{d}{dx} f(x)$$

$$(\%o2) \frac{d}{dx} (f(x) - g(x)) = \frac{d}{dx} f(x) - \frac{d}{dx} g(x)$$

Derivada del producte per una constant i del producte de dues funcions:

```
(%i3) 'diff(a*f(x), x)=diff(a*f(x), x);  
'diff(f(x)*g(x), x)=diff(f(x)*g(x), x);
```

$$(\%o3) \frac{d}{dx} (a f(x)) = a \left(\frac{d}{dx} f(x) \right)$$

$$(\%o4) \frac{d}{dx} (f(x) g(x)) = f(x) \left(\frac{d}{dx} g(x) \right) + g(x) \left(\frac{d}{dx} f(x) \right)$$

Derivada del quocient de dues funcions:

```
(%i5) 'diff(f(x)/g(x), x)=diff(f(x)/g(x), x);
```

$$(\%o5) \frac{\frac{d}{dx} f(x)}{g(x)} = \frac{\frac{d}{dx} f(x)}{g(x)} - \frac{f(x) \left(\frac{d}{dx} g(x) \right)}{g(x)^2}$$

S'observa que l'expressió de la derivada del quocient de dues funcions no està escrita en la manera habitual; per aconseguir-la, simplifiquem l'expressió anterior:

```
(%i6) ratsimp(%);
```

$$(\%o6) \frac{\frac{d}{dx} f(x)}{g(x)} = - \frac{f(x) \left(\frac{d}{dx} g(x) \right) - g(x) \left(\frac{d}{dx} f(x) \right)}{g(x)^2}$$

Derivada de la potència d'una funció amb exponent una funció:

```
(%i7) 'diff(f(x)^(g(x)), x)=diff(f(x)^(g(x)), x);
```

$$(\%o7) \frac{d}{dx} f(x)^{g(x)} = f(x)^{g(x)} \left(\log(f(x)) \left(\frac{d}{dx} g(x) \right) + \frac{g(x) \left(\frac{d}{dx} f(x) \right)}{f(x)} \right)$$

Aquesta expressió es pot operar per obtenir l'expressió habitual de la derivada d'una potència de funcions:

```
(%i8) expand(%);
```

$$(\%o8) \frac{d}{dx} f(x)^{g(x)} = f(x)^{g(x)} \log(f(x)) \left(\frac{d}{dx} g(x) \right) + f(x)^{g(x)-1} g(x) \left(\frac{d}{dx} f(x) \right)$$

Derivada de la composició de dues funcions:

```
(%i9) 'diff(f(g(x)), x)=diff(f(g(x)), x);
```

$$(\%o9) \frac{d}{dx} f(g(x)) = \frac{d}{dx} f(g(x))$$

Es pot observar que la derivada de la funció composta (regla de la cadena) no es desenvolupa com es podia esperar. Tanmateix, no cal deduir que amb wxMaxima no es poden calcular derivades de funcions compostes; es pot fer i en l'exemple que segueix es mostrarà la metodologia per fer-ho.

Vegem un exemple de l'aplicació de les expressions anteriors. Considerem, per exemple, les funcions definides per:

$$f(x) = 1 + x^2, x \in \mathbb{R}; \quad g(x) = \sin(2x+1), x \in \mathbb{R}$$

```
(%i10) f(x):=1+x^2;
      g(x):=sin(2*x-1);
```

```
(%o10) f(x):= 1 + x^2
```

```
(%o11) g(x):= sin(2 x - 1)
```

La derivada de la suma i de la diferència d'aquestes funcions és:

```
(%i12) 'diff(f(x)+g(x), x)=diff(f(x)+g(x), x);
```

$$(\%o12) \frac{d}{dx} (\sin(2x-1) + x^2 + 1) = 2 \cos(2x-1) + 2x$$

```
(%i13) 'diff(f(x)-g(x), x)=diff(f(x)-g(x), x);
```

$$(\%o13) \frac{d}{dx} (-\sin(2x-1) + x^2 + 1) = 2x - 2 \cos(2x-1)$$

La derivada del producte i del quocient d'aquestes funcions és:

```
(%i14) 'diff(f(x)*g(x), x)=diff(f(x)*g(x), x);
```

$$(\%o14) \frac{d}{dx}((x^2+1)\sin(2x-1)) = 2x\sin(2x-1) + 2(x^2+1)\cos(2x-1)$$

```
(%i15) 'diff(f(x)/g(x), x)=diff(f(x)/g(x), x);
```

$$(\%o15) \frac{d}{dx} \frac{x^2+1}{\sin(2x-1)} = \frac{2x}{\sin(2x-1)} - \frac{2(x^2+1)\cos(2x-1)}{\sin(2x-1)^2}$$

Tal com s'ha comentat abans, simplifiquem l'expressió de la derivada del quocient:

```
(%i16) ratsimp(%);
```

$$(\%o16) \frac{d}{dx} \frac{x^2+1}{\sin(2x-1)} = \frac{2x\sin(2x-1) + (-2x^2-2)\cos(2x-1)}{\sin(2x-1)^2}$$

La derivada de la potència d'aquestes funcions és:

```
(%i17) 'diff(f(x)^(g(x)), x)=diff(f(x)^(g(x)), x);
```

$$(\%o17) \frac{d}{dx} (x^2+1)^{\sin(2x-1)} = (x^2+1)^{\sin(2x-1)} \left(2\cos(2x-1)\log(x^2+1) + \frac{2x\sin(2x-1)}{x^2+1} \right)$$

I amb la simplificació comentada:

```
(%i18) expand(%);
```

$$(\%o18) \frac{d}{dx} (x^2+1)^{\sin(2x-1)} = 2(x^2+1)^{\sin(2x-1)} \cos(2x-1) \log(x^2+1) + 2x(x^2+1)^{\sin(2x-1)-1} \sin(2x-1)$$

La derivada de les funcions compostes $(f \circ g)$ i $(g \circ f)$ és:

```
(%i19) 'diff(f(g(x)), x)=diff(f(g(x)), x);
```

$$(\%o19) \frac{d}{dx} (\sin(2x-1)^2+1) = 4\cos(2x-1)\sin(2x-1)$$

```
(%i20) 'diff(g(f(x)), x)=diff(g(f(x)), x);
```

$$(\%o20) \frac{d}{dx} \sin(2(x^2+1)-1) = 4x\cos(2(x^2+1)-1)$$

Derivades d'ordre superior.

Les derivades d'ordre superior, o derivades successives, es defineixen mitjançant:

$$D^k f(x) = D(D^{k-1} f(x)), \quad k \in \mathbb{N}$$

Aquestes derivades d'ordre superior es calculen de forma senzilla amb wxMaxima: la sintaxi només requereix posar després de la variable un nombre natural que indica el nombre de vegades que es deriva la funció.

Per exemple, si es considera la funció polinòmica $P(x) = 4x^3 - 5x^2 + 2x - 6$, la derivada (de primer ordre) es pot calcular mitjançant:

```
(%i1) P1(x):=4*x^3-5*x^2+2*x-6;
(%o1) P1(x):= 4 x^3 - 5 x^2 + 2 x - 6
(%i2) 'diff(P1(x), x, 1)=diff(P1(x), x, 1);
(%o2)  $\frac{d}{dx}(4x^3 - 5x^2 + 2x - 6) = 12x^2 - 10x + 2$ 
(%i3) 'diff(P1(x), x)=diff(P1(x), x);
(%o3)  $\frac{d}{dx}(4x^3 - 5x^2 + 2x - 6) = 12x^2 - 10x + 2$ 
```

S'observa doncs que si després de la variable no es posa cap indicació, per defecte es calcula la derivada primera (o d'ordre 1). Si es vol escriure la derivada amb la notació de l'operador derivació, aleshores es pot fer el següent:

```
(%i4) DP1(x)=diff(P1(x), x);
(%o4) DP1(x)= 12 x^2 - 10 x + 2
```

Per a les derivades d'ordre superior 2, 3 i 4 (qualsevol altra derivada d'ordre superior a 4 és nul·la), la sortida del programa és:

```
(%i5) 'diff(P1(x), x, 2)=diff(P1(x), x, 2);
(%o5)  $\frac{d^2}{dx^2}(4x^3 - 5x^2 + 2x - 6) = 24x - 10$ 
(%i6) 'diff(P1(x), x, 3)=diff(P1(x), x, 3);
(%o6)  $\frac{d^3}{dx^3}(4x^3 - 5x^2 + 2x - 6) = 24$ 
(%i7) 'diff(P1(x), x, 4)=diff(P1(x), x, 4);
(%o7)  $\frac{d^4}{dx^4}(4x^3 - 5x^2 + 2x - 6) = 0$ 
```

Fent servir la notació de l'operador derivació:

```
(%i8) D2P1(x)=diff(P1(x), x, 2);
(%o8) D2P1(x)= 24 x - 10
(%i9) D3P1(x)=diff(P1(x), x, 3);
(%o9) D3P1(x)= 24
```

```
(%i10) D4P1(x)=diff(P1(x), x, 4);  
(%o10) D4P1(x)=0
```

Per poder calcular derivades d'ordre superior aplicant la derivació a la funció derivada d'ordre anterior, cal aplicar la instrucció “define” per poder tenir definida la derivada funcionalment i que el programa pugui treballar amb aquesta funció. Així, per exemple, en el cas del polinomi anterior:

```
(%i11) define (DP1(x) , diff(P1(x), x));  
(%o11) DP1(x):= 12 x2- 10 x + 2  
(%i12) define (D2P1(x) , diff(P1(x), x, 2));  
(%o12) D2P1(x):= 24 x- 10  
(%i13) define (D3P1(x) , diff(P1(x), x, 3));  
(%o13) D3P1(x):= 24  
(%i14) define (D4P1(x) , diff(P1(x), x, 4));  
(%o14) D4P1(x):= 0
```

07-3. Diferencial d'una funció. Aplicacions

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_07-3.wxm**.

Si $f : A \subset \mathbb{R} \rightarrow \mathbb{R}$ és una funció real de variable real diferenciable en un punt $a \in A$ interior del camp d'existència de la funció, s'anomena diferencial de f en el punt a , designada df_a , l'aplicació lineal de $\mathbb{R}$ en $\mathbb{R}$ definida per

$$df_a(h) = Df(a)h, \quad h \in \mathbb{R}$$

Per exemple, es considera la funció diferenciable

$$f(x) = x^2 - 2x + 2, \quad x \in \mathbb{R}$$

Definim-la amb wxMaxima i calculem la seva funció derivada:

```
(%i1) f(x):=x^2-2*x+2;  
(%o1) f(x):=x^2-2 x+2  
(%i2) define (Df(x), diff(f(x), x) );  
(%o2) Df(x):= 2 x-2
```

Aleshores, la diferencial d'aquesta funció en un punt $a \in \mathbb{R}$ és:

```
(%i3) define (dfa(h), Df(a) * h);  
(%o3) dfa(h):=(2 a-2) h
```

La diferencial és l'única aplicació lineal que aproxima localment els increments de la funció en un entorn del punt, és a dir:

$$df_a(h) \approx \Delta f(a; h) = f(a+h) - f(a)$$

En el cas anterior l'increment en un punt és:

```
(%i4) Incr_m_f_a(h)=f(a+h)-f(a);  
(%o4) Incr_m_f_a(h)=(h+a)^2-2 (h+a)-a^2+2 a  
(%i5) expand(%);  
(%o5) Incr_m_f_a(h)=h^2+2 a h-2 h  
(%i6) Incr_m_f_a(h):=h^2+2*a*h-2*h;  
(%o6) Incr_m_f_a(h):=h^2+2 a h+(-2) h
```

La propietat esmentada de la diferencial s'expressa aleshores:

```
(%i6) Incrém_f_a(h):=h^2+2*a*h-2*h;
(%o6) Incrém_f_a(h):=h^2+2 a h+(-2)h
(%i7) '(Incrém_f_a(h)-dfa(h))=Incrém_f_a(h)-dfa(h);
(%o7) Incrém_f_a(h)-dfa(h)=h^2-(2 a-2)h+2 a h-2 h
(%i8) expand(%);
(%o8) Incrém_f_a(h)-dfa(h)=h^2
```

La propietat essencial de la diferencial s'ha aplicat clàssicament a càlculs aproximats que l'existència actualment d'eines potents de càlcul ha deixat gairebé obsolets. Tanmateix, farem tot seguit un exemple d'aplicació de la diferencial d'una funció que en ocasions pot tenir interès.

Es tracta d'establir el subconjunt de punts de la recta real en el qual es pot aproximar l'expressió $(x+2.15)^{1/3}$ per $x^{1/3}$ amb un error inferior a 10^{-4} . Si es considera la funció $f(x) = x^{1/3}$, diferenciable en tota la recta real exceptuant l'origen, aleshores es compleix

```
(%i1) f(x):=x^(1/3);
(%o1) f(x):=x^1/3
(%i2) define (Df(x) , diff(f(x), x) );
(%o2) Df(x):=1/(3 x^2/3)
```

Per tant la funció diferencial està definida mitjançant

```
(%i3) define (dfa(h), Df(a) * h);
(%o3) dfa(h):=h/(3 a^2/3)
```

El valor de la diferencial en $h = 2.15$ és una aproximació de l'increment degut a aquest desplaçament, és a dir:

```
(%i4) dfa(2.15);
(%o4) 0.716666666666667/a^2/3
```

Per tant, es tracta de trobar els punts que fan que aquest increment no superi la fita fixada, fet que s'expressa mitjançant la desigualtat:

```
(%i5) 0.716666666666667/a^(2/3)<10^(-4);
(%o5) 0.716666666666667/a^2/3 < 1/10000
```

Resolent aquesta inequació s'obté:

$$\begin{aligned} &(\%i6) \ a > (0.71666666666667 * 10000)^{(3/2)}; \\ &(\%o6) \ a > 606702.5325173516 \end{aligned}$$

Així doncs, la condició es compleix en l'interval $]606702.53252, +\infty[$.

07-4. Funcions implícites. Derivació de funcions implícites


Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_07-4.wxm**.

Les funcions implícites d'una variable o funcions definides implícitament són aquelles que es defineixen mitjançant equacions de corbes planes en les quals intervenen dues variables. Una corba plana és un conjunt de punts del pla definits per una equació del tipus $F(x, y) = 0$. Mitjançant aquesta equació es pot definir implícitament una funció $x \mapsto y(x)$ tal que seva la imatge en un punt x és el valor $y(x)$ tal que $F(x, y(x)) = 0$. Per exemple, la corba plana d'equació $x^3 + y^3 = 1$ defineix implícitament una funció $x \mapsto y(x)$ tal que la imatge és el valor tal que $x^3 + y(x)^3 = 1$. Els detalls sobre aquest tema i les propietats de les funcions definides implícitament, es poden trobar a les referències.

Per representar gràficament amb wxMaxima una corba plana cal carregar primer el paquet anomenat “draw2d” i fer servir la instrucció “implicit”, especificant l'equació de la corba i els rangs de les variables, tal com es veurà a continuació. El gràfic resultant es pot modificar amb les opcions corresponents, fent servir el quadre de diàleg a l'efecte. Si es vol obtenir la gràfica “enganxada” al mateix fitxer de càlculs, aleshores cal fer servir la instrucció “wxdraw2d”.


Així, la representació gràfica de la corba plana $x^3 + y^3 = 1$ s'obté de la manera següent:

```
(%i1) load(draw)$  
      wxdraw2d(  
        grid = true, line_type = solid, line_width = 2, color = blue,  
        implicit(x^3+y^3-1=0, x, -3,3, y, -3,3));
```



Vegem un altre exemple en el qual es canvia el color emprat en la gràfica de la corba plana; es tracta de la corba plana d'equació $x^3 - xy + y^3 = 1$:

```
(%i3) wxdraw2d(
      grid = true, line_type = solid, line_width = 2, color = red,
      implicit(x^3-x*y+y^3-1=0, x, -2,2, y, -2,2));
```



Derivació implícita.

Si una corba plana defineix implícitament una funció diferenciable, es poden calcular les derivades de la funció definida implícitament. Vegem tot seguit amb un exemple la metodologia i la sintaxi que cal aplicar. Considerem l'equació $x^3 + y^3 - xy - 1 = 0$, la qual defineix una corba plana representada a la Figura 07-3(b).

En primer lloc cal escriure l'equació de referència mitjançant la qual es relacionen inicialment les variables:

```
(%i1) Eq1:x^3-x*y+y^3-1=0;
(%o1) y^3-x y+x^3-1=0
```

A continuació cal especificar quina de les variables és la que es considera funció definida implícitament i quina és la que segueix sent la variable. La instrucció de wxMaxima és “depends” amb un parèntesi en el qual es posen las variables ordenades: en primer lloc la “funció” i en segon lloc la “variable”; en aquest cas la sintaxi és

```
(%i2) depends(y,x);
(%o2) [y(x)]
```

A partir de l'aplicació d'aquesta instrucció, el programa interpreta que y és una funció de x definida mitjançant una equació. Si la funció definida implícitament és diferenciable, es pot calcular la derivada d'aquesta funció derivant en l'equació inicial, mitjançant la instrucció:

```
(%i3) diff(Eq1,x);
```

$$(\%o3) 3 y^2 \left(\frac{d}{dx} y \right) - x \left(\frac{d}{dx} y \right) - y + 3 x^2 = 0$$

En aquesta equació es pot veure com wxMaxima interpreta que s'ha assignat el rol funcional $y(x)$ i s'ha derivat l'equació inicial considerant x la variable i y una funció de x . Es pot designar l'equació anterior amb una referència, per exemple

```
(%i4) 'Eq2=Eq2:%;
```

$$(\%o4) Eq2 = \left(3 y^2 \left(\frac{d}{dx} y \right) - x \left(\frac{d}{dx} y \right) - y + 3 x^2 = 0 \right)$$

Aquesta equació es pot reescriure fent servir la notació de la derivada amb l'operador derivació:

```
(%i5) Eq2b:3*y^2*Dy(x)-x*Dy(x)-y+3*x^2=0;
```

$$(\%o5) 3 Dy(x) y^2 - y - x Dy(x) + 3 x^2 = 0$$

Si es vol calcular l'equació de la recta tangent a la corba definida per l'equació Eq1 en un punt, per exemple el punt (1,0), cal substituir aquests valors en l'equació Eq2, obtenint-se:

```
(%i6) Eq3:-1*(diff(y,x,1))-0+3*1^2=0;
```

$$(\%o6) 3 - \frac{d}{dx} y = 0$$

Per tant es compleix

```
(%i7) 'Dy(1)=3;
```

$$(\%o7) Dy(1) = 3$$

L'equació de la recta tangent buscada és doncs

```
(%i8) y-0=3*(x-1);
```

$$(\%o8) y = 3(x-1)$$

Si es vol calcular la derivada segona de la funció definida implícitament, cal derivar implícitament en l'equació Eq2, i s'obté

```
(%i9) diff(Eq2,x);
(%o9) 3 y^2  $\left(\frac{d^2}{dx^2} y\right) - x \left(\frac{d^2}{dx^2} y\right) + 6 y \left(\frac{d}{dx} y\right)^2 - 2 \left(\frac{d}{dx} y\right) + 6 x = 0$ 
(%i10) Eq4: %;
(%o10) 3 y^2  $\left(\frac{d^2}{dx^2} y\right) - x \left(\frac{d^2}{dx^2} y\right) + 6 y \left(\frac{d}{dx} y\right)^2 - 2 \left(\frac{d}{dx} y\right) + 6 x = 0$ 
```

Per calcular el valor de la derivada segona en un punt, per exemple en $x=1$, cal substituir ara els valors $x=1, y(1)=0, Dy(1)=3$ resulta:

```
(%i11) Eq5: -(diff(y,x,2))-2*3+6=0;
(%o11)  $-\frac{d^2}{dx^2} y = 0$ 
```

Finalment doncs:

```
(%i12) 'D2y(1)=0;
(%o12) D2y(1)=0
```